HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Mechanical Engineering



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	**Keystrokes**	**Display**
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^T SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^T THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^T R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

1. GEAR FORCES 1

Computes the forces on helical, bevel, or worm gears.

2. STRESS ON AN ELEMENT 9

Performs Mohr circle stress analysis, and reduces rosette strain gage data.

3. EQUATIONS OF STATE 19

Solves both the ideal gas and Redlich-Kwong equations of state.

4. SODERBERG'S EQUATION FOR FATIGUE 27

Given six variables, this program will calculate the seventh unknown.

5. SPRING CONSTANT 34

Solves any general linear equation of the form $(y - y_o) = m(x - x_o)$.

6. PROGRESSION OF A SLIDER CRANK 39

Calculates displacement, velocity, acceleration, and other quantities.

7. FREE VIBRATIONS 48

Solves the differential equation $m\ddot{x} + c\dot{x} + kx = 0$.

8. INTERFERENCE FITS 56

Calculates contact pressure or interference for concentric cylinders.

9. LINEAR OF ANGULAR DEFORMATION 64

Solves for linear deflection under tensile load or angular deflection under torque.

10. CONSTANT ACCELERATION 69

Calculates displacement, acceleration, initial velocity and time for an object undergoing constant acceleration.

GEAR FORCES

This program computes three mutually perpendicular forces resulting from input torque on helical, bevel, or worm gears.

Helical gear equations:

$$F_t = \frac{T}{r}$$

$$F_{gs} = F_t \tan \phi$$

$$F_{gax} = F_t \tan \alpha$$

$$\tan \phi = \frac{\tan \phi_n}{\cos \alpha}$$



Figure 1-Helical Gear

where:

T is the input torque;

r is the pitch radius of the input gear;

F_t is the tangential force;

α is the helix angle measured from the axis of the gear (for spur gears $\alpha = 0$);

ϕ_n is the pressure angle measured perpendicular to the gear tooth;

ϕ is the pressure angle measured perpendicular to the gear axis;

F_{gs} is the radial force trying to separate the gears;

F_{gax} is the force parallel to the gear axis.

Bevel gear equations:

$$F_t = \frac{T}{r}$$

$$F_{bpax} = F_t \left(\frac{\tan \phi_n \sin (\text{cone}\angle)}{\cos \alpha} + \tan \alpha \cos (\text{cone}\angle) \right)$$

$$F_{bgax} = F_t \left(\frac{\tan \phi_n \cos (\text{cone}\angle)}{\cos \alpha} - \tan \alpha \sin (\text{cone}\angle) \right)$$

$$\tan \phi = \frac{\tan \phi_n}{\cos \alpha}$$



Figure 2—Spiral Bevel Gears

where:

- T is the input (pinion) torque;
- r is the pitch radius of the pinion gear;
- F_t is the tangential force;
- α is the pinion spiral angle (zero for straight tooth bevel gears);
- ϕ_n is the pressure angle measured perpendicular to the gear tooth;
- ϕ is the pressure angle measured perpendicular to the gear axis;
- Cone$\angle$ is the pitch cone angle of the pinion;
- F_{bpax} is the force along the axis of the bevel pinion;
- F_{bgax} is the force along the axis of the bevel gear.

Worm gear equations:

$$F_t = \frac{T}{r}$$

$$F_{ws} = F_t \left(\frac{\sin \phi_n}{\cos \phi_n \sin \alpha + f \cos \alpha} \right)$$

$$F_{gax} = F_t \frac{1 - \frac{f \tan \alpha}{\cos \phi_n}}{\tan \alpha + \frac{f}{\cos \phi_n}}$$

$$\tan \phi = \frac{\tan \phi_n}{\cos \alpha}$$



Figure 3
WORM GEAR

where:

T is the input (worm) torque;

n is the pitch radius of the worm;

F_t is the tangential force on the worm;

α is the lead angle of the worm ($\alpha = \tan^{-1}(L/2\pi r)$, where L is the lead of the worm);

ϕ_n is the pressure angle measured perpendicular to the worm teeth;

ϕ is the pressure angle measured parallel to the worm axis;

f is the coefficient of friction;

F_{ws} is the separating force between the worm and gear;

F_{gax} is the force parallel to the gear axis.

Example 1:

A helical gear with pitch radius 12 cm has a torque applied to it of 450,000 dyne-cm. The helix angle is 30° and the normal pressure angle, measured perpendicular to a tooth is 17.5°. Find the tangential, separating, and thrust forces.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
[XEQ] [ALPHA] GEAR [ALPHA]	T?
450000 [R/S]	R?
12 [R/S]	F=37,500.00
[R/S]	ALPHA?
30 [R/S]	AN?
17.5 [R/S]	
[XEQ] [ALPHA] HEL [ALPHA]	FGS=13,652.84
[R/S]	FGAX=21,650.64

Example 2:

A spiral pinion with mean radius 1.73 inches is subjected to a torque of 745 in-lb. The pinion is cut with a normal pressure angle of 20°, a spiral angle of 35°, and a pitch cone of 18°. Find the forces acting on the pinion. Rotation is in the direction of the concave side of the pinion teeth, so α is positive 35°.

Keystrokes:	Display:
[XEQ] [ALPHA] GEAR [ALPHA]	T?
745 [R/S]	R?
1.73 [R/S]	F=430.64
[R/S]	ALPHA?
35 [R/S]	AN?
20 [R/S]	
[XEQ] [ALPHA] BEV [ALPHA]	CONE∡?
18 [R/S]	FBPAX=345.90
[R/S]	FBGAX=88.80

User Instructions

				SIZE: 013
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] GEAR	T?
3	Input values:			
	Torque	T	[R/S]	R?
	Pitch radius	r	[R/S]	$F=(F_t)$
			[R/S]	ALPHA?
4	Input helix angle for helical gears, or			
	spiral angle for spiral bevel gears, or			
	lead angle for worm gears	α	[R/S]	AN?
5	Input normal pressure angle	ϕ_n	[R/S]	
	OR			
	Input pressure angle	ϕ	[XEQ] N	
6	Execute appropriate program			
	For helical gears		[XEQ] HEL	FGS=
			[R/S]	FGAX=
	For bevel gears		[XEQ] BEV	CONE∠?
	input cone angle	CONE ∠	[R/S]	FBPAX=
			[R/S]	FBGAX=
	For worm gears		[XEQ] WORM	F?
	input coefficient of friction	f	[R/S]	FWS=
			[R/S]	FGAX=

Program Listings

Initialization

```
 01♦LBL "GEA
R"
 02 FIX 2
 03 "T?"
 04 PROMPT
 05 "R?"
 06 PROMPT
 07 /
 08 STO 06
 09 "F="
 10 XEQ 05
 11 "ALPHA?"
 12 PROMPT
 13 STO 11
 14 "AN?"
 15 PROMPT
 16♦LBL 10
 17 STO 05
 18 STOP
```

Helical gears

```
 19♦LBL "HEL
"
 20 RCL 06
 21 RCL 11
 22 TAN
 23 *
 24 RCL 05
 25 TAN
 26 RCL 11
 27 COS
 28 /
 29 RCL 06
 30 *
 31 "FGS="
 32 XEQ 05
 33 X<>Y
 34 "FGAX="
 35 GTO 05
```

Bevel gears

```
 36♦LBL "BEV
"
 37 "CONE∡?"
 38 PROMPT
 39 STO 12
 40 RCL 12
 41 RCL 05
 42 TAN
 43 RCL 11
 44 COS
 45 /
 46 RCL 06
 47 *
 48 P-R
 49 RCL 06
 50 RCL 11
 51 TAN
 52 *
 53 RCL 12
 54 X<>Y
 55 P-R
 56 R↑
 57 +
 58 RDN
 59 -
 60 R↑
 61 "FBPAX="
 62 XEQ 05
 63 X<>Y
 64 "FBGAX="
 65 GTO 05
```

Worm gears

```
 66♦LBL "WOR
M"
 67 "F?"
 68 PROMPT
 69 STO 04
 70 RCL 05
 71 SIN
 72 LASTX
 73 COS
 74 RCL 11
 75 SIN
 76 *
 77 RCL 11
 78 COS
 79 RCL 04
 80 *
 81 +
 82 /
 83 RCL 06
 84 *
 85 "FWS="
 86 XEQ 05
 87 1
 88 RCL 11
 89 TAN
 90 RCL 04
 91 *
 92 RCL 05
 93 COS
 94 /
 95 -
 96 RCL 11
 97 TAN
 98 RCL 04
 99 RCL 05
```

Program Listings

100 COS	
101 /	
102 +	
103 /	
104 RCL 06	
105 *	
106 "FGAX="	
107♦LBL 05	Display
108 ARCL X	
109 AVIEW	
110 STOP	
111 RTN	
112♦LBL "N"	Calculate Θn from Θ
113 TAN	
114 RCL 11	
115 COS	
116 *	
117 ATAN	
118 GTO 10	
119 .END.	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	f		
05	ϕn	55	
	F_t		
10		60	
	α		
	Cone		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 43 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

STRESS ON AN ELEMENT

This program reduces data from rosette strain gage measurements and/or performs Mohr circle stress analysis calculations.

Correlations for rectangular and equiangular rosette configurations are included.

Strain Gage Equations:

CONFIGURATION CODE	1	2
TYPE OF ROSETTE	RECTANGULAR	DELTA (EQUIANGULAR)
		
PRINCIPAL STRAINS: ϵ_1, ϵ_2	$\frac{1}{2}\left[\epsilon_a + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}\,\right]$	$\frac{1}{3}\left[\epsilon_a + \epsilon_b + \epsilon_c \pm \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}\,\right]$
CENTER OF MOHR CIRCLE: $\frac{s_1 + s_2}{2}$	$\frac{E(\epsilon_a + \epsilon_c)}{2(1 - \nu)}$	$\frac{E(\epsilon_a + \epsilon_b + \epsilon_c)}{3(1 - \nu)}$
MAXIMUM SHEAR STRESS: τ_{max}	$\frac{E}{2(1 + \nu)} \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2}$	$\frac{E}{3(1 + \nu)} \sqrt{2(\epsilon_a - \epsilon_b)^2 + 2(\epsilon_b - \epsilon_c)^2 + 2(\epsilon_c - \epsilon_a)^2}$
ORIENTATION OF PRINCIPAL STRESSES	$\tan^{-1}\left[\frac{2\epsilon_b - \epsilon_a - \epsilon_c}{\epsilon_a - \epsilon_c}\right]$	$\tan^{-1}\left[\frac{\sqrt{3}(\epsilon_c - \epsilon_b)}{(2\epsilon_a - \epsilon_b - \epsilon_c)}\right]$

The Mohr circle portion of the program converts an arbitrary stress configuration to principal stresses, maximum shear stress and rotation angle. It is then possible to calculate the state of stress for an arbitrary orientation Θ´.



Mohr Circle Equations:

$$\tau_{max} = \sqrt{\left(\frac{S_x - S_y}{2}\right)^2 + \tau_{xy}^{\,2}}$$

$$S_1 = \frac{S_x + S_y}{2} + \tau_{max}$$

$$S_2 = \frac{S_x + S_y}{2} - \tau_{max}$$

$$\theta = \tfrac{1}{2}\tan^{-1}\left(\frac{2\tau_{xy}}{S_x - S_y}\right)$$

$$s = \frac{s_1 + s_2}{2} + \tau_{max} \cos 2\theta'$$

$$\tau = \tau_{max} \sin 2\theta'$$

where:

s is the normal stress, and τ is the shear stress.

ε_a, ε_b, and ε_c are the strains measured using rosette gages;

S_x is the stress in the x direction for Mohr circle input;

S_y is the stress in the y direction for Mohr circle input;

τ_{xy} is the shear stress on the element for Mohr circle input;

ε_1 and ε_2 are the principal strains;

S_1 and S_2 are the principal normal stresses;

τ_{max} is the maximum shear stress;

θ is the counterclockwise angle of rotation from the specified axis to the principal axis. Note that this is opposite to the normal Mohr circle convention.

θ' is an arbitrary rotation angle from the original (x,y) axis;

E is modulus of elasticity.

Reference:

Spotts, M.F., Design of Machine Elements; Prentice-Hall, 1971.

Beckwith, T. G., Buck, N. L., Mechanical Measurements; Addison-Wesley, 1969.

Remarks:

Negative stresses and strains indicate compression. Positive and negative shear are represented below:



Example:

A rectangular rosette measures the strains below. What are the principal strains and principal stresses?

$\varepsilon_a = 90 \times 10^{-6}$ $\varepsilon_b = 137 \times 10^{-6}$ $\varepsilon_c = 305 \times 10^{-6}$

$\nu = 0.3$ $E = 30 \times 10^6$ psi

Keystrokes:	Display:
[USER]	
[XEQ[[ALPHA] SIZE [ALPHA] 016	
[XEQ] [ALPHA] ROSETTA [ALPHA]	RECT? Y/N
Y [R/S]	E?
30 [EEX] 6 [R/S]	RATIO?
.3 [R/S]	EA?
90 [EEX] 6 [CHS] [R/S]	EB?
137 [EEX] 6 [CHS] [R/S]	EC?
305 [EEX] 6 [CHS] [R/S]	E=320.9E-6
[R/S]	E=74.14E-6
[R/S]	∠=14.69E0
[R/S]	S=11.31E3
[R/S]	S=5.618E3
[R/S]	TMAX=2.847E3
[R/S]	∠=14.69E0

User Instructions

				SIZE: 016
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode.		[USER]	
2	For Mohr's circle go to step 9.			
3	Initialize for rosetta strain gage data.		[XEQ] ROSETTA	RECT? Y/N
4	Choose configuration:			
	rectangular OR	Y	[R/S]	E?
	equiangular	N	[R/S]	E?
5	Input modulus of elasticity.	E	[R/S]	RATIO?
6	Input poisson's ratio.	ν	[R/S]	EA?
7	Input strains and calculate principal	ε_a	[R/S]	EB?
	strains and rotation angle.	ε_b	[R/S]	EC?
		ε_c	[R/S]	E=(ε_1)
			[R/S]	E=(ε_2)
			[R/S]	∡=(θ)
8	Calculate Mohr's circle data from strain			
	gage data.		[R/S]	S=(S_1)
			[R/S]	S=(S_2)
			[R/S]	TMAX=
			[R/S]	∡=(θ)
9	Initialize for Mohr's circle.		[XEQ] MOHR	SX?
10	Input stresses and calculate principal	s_x	[R/S]	SY?
	stresses and rotation angle.	s_y	[R/S]	TXY?
		τ_{xy}	[R/S]	S=(S_1)
			[R/S]	S=(S_2)
			[R/S]	TMAX=
			[R/S]	∡=(θ)
11	Optional: calculate stress at specific			
	angle.		[E]	∡?

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
12	Key in angle.	θ	[R/S]	S=
			[R/S]	T=

Program Listings

Initialization

```
 01♦LBL "ROS
ETTA"
 02 ENG 3
 03 1
 04 STO 10
 05 "RECT? Y
/N"
 06 AON
 07 PROMPT
 08 AOFF
 09 ASTO X
 10 "N"
 11 ASTO Y
 12 X=Y?
 13 ISG 10
 14♦LBL 00
 15 "E?"
 16 PROMPT
 17 STO 15
 18 "RATIO?"
 19 PROMPT
 20 STO 09
 21 "EA?"
 22 PROMPT
 23 STO 11
 24 "EB?"
 25 PROMPT
 26 STO 12
 27 "EC?"
 28 PROMPT
 29 STO 13
 30 RCL 11
```

Calculate ε_1, ε_2

```
 31 GTO IND
10
 32♦LBL 02
 33 RCL 12
 34 +
 35♦LBL 01
 36 RCL 13
 37 +
 38 STO 06
 39 0
 40 GTO IND
10
 41♦LBL 02
 42 RCL 13
 43 RCL 11
 44 -
 45♦LBL 01
 46 RCL 12
 47 RCL 13
 48 -
 49 R-P
 50 RCL 11
 51 RCL 12
 52 -
 53 R-P
 54 2
 55 SQRT
 56 *
 57 STO 05
 58 2
 59 GTO IND
10
 60♦LBL 02
 61 1
 62 +
 63♦LBL 01
 64 ST/ 05
 65 ST/ 06
 66 "E="
 67 ASTO 00
 68 XEQ 05
```

Calculate τ_{max} and $\frac{s_1 + s_2}{2}$ from strains

```
 69 RCL 15
 70 RCL 09
 71 1
 72 +
 73 /
 74 ST* 05
 75 RCL 15
 76 1
 77 RCL 09
 78 -
 79 /
 80 ST* 06
 81 RCL 13
 82 RCL 12
 83 -
 84 3
 85 SQRT
 86 GTO IND
10
 87♦LBL 01
 88 2
 89 RCL 12
 90 *
 91 RCL 11
 92 -
 93 RCL 13
 94 -
 95 RCL 11
 96 RCL 13
 97 GTO 04
 98♦LBL 02
```

Program Listings

```
 99 *
100 2
101 RCL 11
102 *
103 RCL 12
104 -
105 RCL 13
106♦LBL 04
107 -
108 XEQ 06
109 RDN
110 "∡="
111 ARCL X
112 AVIEW
113 STOP
114 GTO D
```

Calculate τ_{max} and $\frac{s_1 + s_2}{2}$ from s_x, s_y, τ_{xy}

```
115♦LBL "MOH
R"
116 "SX?"
117 PROMPT
118 "SY?"
119 PROMPT
120 "TXY?"
121 PROMPT
122 R↑
123 R↑
124 STO 03
125 STO 06
126 R↑
127 ST+ 06
128 -
129 STO 04
130 2
131 ST/ 06
132 /
133 R↑
134 STO 02
135 ST+ 02
136 R-P
137 STO 05
138 RCL 02
139 CHS
140 RCL 04
141 XEQ 06
142 GTO D
143♦LBL 06
144 X≠0?
145 /
146 ATAN
147 STO 02
148 2
149 /
150 0
151 RTN
```

Output S_1, S_2, τ_{max} and θ'

```
152♦LBL D
153 "S="
154 ASTO 00
155 XEQ 05
156 RCL 05
157 "TMAX="
158 ARCL X
159 AVIEW
160 STOP
```

Calculate ε_1, ε_2, or S_1, S_2

```
161 RCL 02
162 2
163 /
164 "∡="
165 ARCL X
166 AVIEW
167 STOP
168♦LBL E
169 "∡?"
170 PROMPT
171 ENTER↑
172 +
173 RCL 02
174 -
175 RCL 05
176 P-R
177 RCL 06
178 +
179 "S="
180 ARCL X
181 AVIEW
182 STOP
183 X<>Y
184 "T="
185 ARCL X
186 AVIEW
187 STOP
```

```
188♦LBL 05
189 RCL 06
190 RCL 05
191 +
192 ARCL X
193 AVIEW
194 STOP
195 RCL 06
196 RCL 05
197 -
198 CLA
199 ARCL 00
200 ARCL X
201 AVIEW
```

Program Listings

```
202 STOP
203 RTN
204 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Temp: ALPHA Storage	50	
	2θ		
	Sx		
	Sx-Sy		
05	τ_{max}	55	
	$(S_1+S_2)/2$		
	ν		
10	Control	60	
	ε_a		
	ε_b		
	ε_c		
15	E	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 61 USER MODE

ENG 3 FIX ___ SCI ___ ON X OFF ___

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EQUATIONS OF STATE

This program provides both ideal gas and Redlich-Kwong equations of state. Given four of the five state variables, the fifth is calculated. For the Redlich-Kwong solution, the critical pressure and temperature of the gas must be known. They are not needed for ideal gas solutions.

Values of the Universal Gas Constants

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.7302	atm-ft^3/lb mole-^oR	atm	ft^3/lb mole	^oR
10.73	psi-ft^3/lb mole-^oR	psi	ft^3/lb mole	^oR
1545	psf-ft^3/lb mole-^oR	psf	ft^3/lb mole	^oR

Critical Temperatures and Pressures

Substance	T_c, K	T_c, ^oR	P_c, ATM
Ammonia	405.6	730.1	112.5
Argon	151	272	48.0
Carbon dioxide	304.2	547.6	72.9
Carbon monoxide	133	239	34.5
Chlorine	417	751	76.1
Helium	5.3	9.5	2.26
Hydrogen	33.3	59.9	12.8
Nitrogen	126.2	227.2	33.5
Oxygen	154.8	278.6	50.1
Water	647.3	1165.1	218.2
Dichlorodifluoromethane	384.7	692.5	39.6
Dichlorofluoromethane	451.7	813.1	51.0
Ethane	305.5	549.9	48.2
Ethanol	516.3	929.3	63
Methanol	513.2	923.8	78.5
n-Butane	425.2	765.4	37.5
n-Hexane	507.9	914.2	29.9
n-Pentane	469.5	845.1	33.3
n-Octane	568.6	1023.5	24.6
Trichlorofluoromethane	471.2	848.1	43.2

Equations:

Ideal gas: $PV = nRT$

Redlich-Kwong:

$$P = \frac{nRT}{(V - b)} - \frac{a}{T^{1/2}\, V\, (V + b)}$$

$$a = 4.934\ b\ nRT_c^{1.5}$$

$$b = 0.0867\ \frac{nRT_c}{P_c}$$

where:

P is the absolute pressure;

V is the volume;

n is the number of moles present;

R is the universal gas constant;

T is the absolute temperature;

T_c is the critical temperature;

P_c is the critical pressure.

Remarks:

P, V, n and T must have units compatible with R.

At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.

No equation of state is valid for all substances over an infinite range of conditions. The Redlich-Kwong equation gives moderate to good accuracy for a variety of substances over a wide range of conditions. Results should be used with caution and tempered by experience.

Solutions for V, n, R and T, using the Redlich-Kwong equation, require an iterative technique. Newton's method is employed using the ideal gas law to generate the initial guess. Iteration time is generally a function of the amount of deviation from ideal gas behavior. For extreme cases, the routine may fail to converge entirely, resulting in a "DATA ERROR".

Example 1:

0.63 g moles of air are enclosed in a 25,000 cm^3 space at 1200 K. What is the pressure in bars? Assume an ideal gas.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] ID [ALPHA]	P?
0 [R/S]	V?
25,000 [R/S]	N?
0.63 [R/S]	R?
83.14 [R/S]	T?
1200 [R/S]	P=2.52

Example 2:

The specific volume of a gas in a container is 800 cm^3/g mole. The temperature will reach 400K. What will the pressure be, according to the Redlich-Kwong relation?

P_c = 48.2 atm

T_c = 305.5 K

R = 82.05 cm^3 - atm/g mole-K

Keystrokes:	Display:
[XEQ] [ALPHA] RK [ALPHA]	TC?
305.5 [R/S]	PC?
48.2 [R/S]	P?
0 [R/S]	V?
800 [R/S]	N?
1 [R/S]	R?
82.05 [R/S]	T?
400 [R/S]	P=36.27

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Select equation:			
	a) Ideal gas OR		[XEQ] ID	P?
	b) Redlich-Kwong		[XEQ] RK	TC?
		Tc	[R/S]	PC?
		Pc	[R/S]	P?
3	Input known values:	P	[R/S]	V?
	(Enter zero for the unknown values)	V	[R/S]	N?
		n	[R/S]	R?
		R	[R/S]	T?
		T	[R/S]	(unknown) =

Program Listings

Initialization

```
01♦LBL "ID"
02 0
03 SF 00
04 GTO 00
05♦LBL "RK"
06 1
07 CF 00
08 "TC?"
09 PROMPT
10 STO 13
11 "PC?"
12 PROMPT
13 STO 14
14♦LBL 00
15 SF 02
16 CF 01
17 FIX 2
18 "P?"
19 PROMPT
20 5
21 XEQ 00
22 "V?"
23 PROMPT
24 6
25 XEQ 00
26 "N?"
27 PROMPT
28 7
29 XEQ 00
30 "R?"
31 PROMPT
32 8
33 XEQ 00
34 "T?"
35 PROMPT
36 CF 02
37 9
38♦LBL 00
39 CF 01
40 STO 01
41 RDN
42 STO IND
01
43 X≠0?
44 GTO 01
45 R↑
46 STO 10
47 1
48 STO IND
01
49♦LBL 01
50 FS? 02
51 RTN
52 GTO IND
10
```

Calculate unknown

```
53♦LBL 05
54 "P="
55 GTO 00
56♦LBL 06
57 "V="
58♦LBL 00
59 RCL 07
60 RCL 08
61 *
62 RCL 09
63 *
64 RCL 05
65 RCL 06
66 *
67 /
68 STO IND
10
69 GTO 00
70♦LBL 07
71 SF 01
72 "N="
73 GTO 01
74♦LBL 08
75 SF 01
76 "R="
77 GTO 01
78♦LBL 09
79 "T="
80 SF 01
81♦LBL 01
82 RCL 05
83 RCL 06
84 *
85 RCL 07
86 /
87 RCL 08
88 /
89 RCL 09
90 /
91 STO IND
10
92♦LBL 00
93 FS? 00
94 GTO 10
```

If ideal, display

```
95 XEQ 01
96 GTO 00
97♦LBL 02
98 FS? 01
```

Program Listings

```
 99 XEQ 01
100♦LBL 00
101 RCL 00
```

Calculate using Redlich-Kwong equations

```
102 RCL 09
103 *
104 RCL 06
105 RCL 12
106 -
107 STO 04
108 /
109 RCL 11
110 RCL 09
111 SQRT
112 /
113 STO 02
114 RCL 06
115 /
116 LASTX
117 RCL 12
118 +
119 STO 03
120 /
121 -
122 RCL 05
123 -
124 XEQ IND
10
125 /
126 ST- IND
10
127 RCL IND
10
128 /
129 ABS
130 1 E-4
131 X<=Y?
132 GTO 02
133 RCL IND
10
134 GTO 10
```

$\frac{\partial P}{\partial V}$

```
135♦LBL 06
136 RCL 06
137 ENTER↑
138 +
139 RCL 12
140 +
141 RCL 02
142 *
143 RCL 03
144 RCL 06
145 *
146 X↑2
147 /
148 RCL 00
149 RCL 09
150 *
151 RCL 04
152 X↑2
153 /
154 -
155 RTN
```

$\frac{\partial P}{\partial T}$

```
156♦LBL 09
157 RCL 00
158 RCL 04
159 /
160 RCL 02
161 2
162 /
163 RCL 09
164 /
165 RCL 06
166 /
167 RCL 03
168 /
169 +
170 RTN
```

$\frac{\partial P}{\partial n}$ or $\frac{\partial P}{\partial R}$

```
171♦LBL 07
172♦LBL 08
173 RCL 09
174 RCL 06
175 *
176 RCL 04
177 X↑2
178 /
179 RCL 06
180 ENTER↑
181 +
182 RCL 12
183 +
184 RCL 00
185 /
186 RCL 06
187 /
188 RCL 03
189 X↑2
190 /
191 RCL 02
192 *
193 -
194 RCL 00
195 *
196 RCL IND
10
```

Program Listings

```
197 /
198 RTN
199♦LBL 05
200 LASTX
201 +
```

Calculate a, b

```
202 STO 05
203 GTO 10
204♦LBL 01
205 RCL 07
206 RCL 08
207 *
208 STO 00
209 .0867
210 RCL 14
211 /
212 X<>Y
213 RCL 13
214 *
215 *
216 STO 12
217 LASTX
218 *
219 RCL 13
220 SQRT
221 *
222 4.934
223 *
224 STO 11
```

Display

```
225 RTN
226♦LBL 10
227 ARCL X
228 AVIEW
229 STOP
230 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	NR	50	
	Temp storage index		
	$a/T^{1/2}$		
	(V+b)		
	(V-b)		
05	P	55	
	V		
	n		
	R		
	T		
10	control	60	
	a		
	b		
	Tc		
	Pc		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 61 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Ideal	Redlich-Kwong
01		Calculate a,b	Don't calculate a,b
02		Input data	Calculate

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SODERBERG'S EQUATION FOR FATIGUE

This program will calculate the seventh variable from the other six values in Soderberg's equation. It is useful in sizing parts for cyclic loading, calculating factors of safety, choosing materials based on size constraints and estimating the fatigue resistance of available parts. Soderberg's equation is graphically represented in figure 1.

Equations:



Working Stress Diagram
Figure 1

$$\frac{S_{yp}}{FS} = \frac{S_{max} + S_{min}}{2} + K\left(\frac{S_{yp}}{S_e}\right)\left(\frac{(S_{max} - S_{min})}{2}\right)$$

$$\frac{S_{max} + S_{min}}{2} = \frac{P_{max} + P_{min}}{2A}$$

$$\frac{S_{max} - S_{min}}{2} = \frac{P_{max} - P_{min}}{2A}$$

where:

S_{yp} is the yield point stress of the material;

S_e is the material endurance stress from reversed bending tests;

K is the stress concentration factor for the part;

FS is the factor of safety ($FS \geq 1.00$)

S_{max} is the maximum stress;

S_{min} is the minimum stress;

P_{max} is the maximum load;

P_{min} is the minimum load;

A is the cross sectional area of the part over which the force is evenly distributed.

Reference:

Spots, M. F., Design of Machine Elements; Prentice-Hall, Inc., 1971.

Baumeister, T., Marks Standard Handbook for Mechanical Engineers; McGraw-Hill Book Company, 1967.

Remarks:

If S_{max} and S_{min} are to be input or calculated instead of P_{max} or P_{min}, simply use 1.00 for the value of area.

This implementation of Soderberg's equation is for ductile materials only.

Values of stress concentration factors and material endurance limits may be found in the referenced sources.

In the presence of corrosive media, or for rough surfaces, fatigue effects may be much more significant than predicted by this program.

Example:

What is the maximum permissible cyclic load for a part if the minimum load is 2000 pounds and the area is 0.5 square inches?

$$S_{yp} = 70000 \text{ psi}$$

$$S_e = 25000 \text{ psi}$$

$$K = 1.25$$

$$FS = 2.0$$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] SEF [ALPHA]	SYP?
70000 [R/S]	SE?
25000 [R/S]	A?
0.5 [R/S]	K?
1.25 [R/S]	PMAX?
[R/S]	PMIN?
2000 [R/S]	FS?
2.0 [R/S]	
[XEQ] [ALPHA] PMAX [ALPHA]	PMAX=8.889E3

If P_{max} is changed to 10000 pounds, what will S_e have to be?

10000 [STO] 03	
[XEQ] [ALPHA] SE [ALPHA]	SE=30.43E3

User Instructions

				SIZE: 010
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] SEF	SYP?
3	Input values	Syp	[R/S]	SE?
	(skip unknown by pressing [R/S])	Se	[R/S]	A?
		Area*	[R/S]	K?
		K	[R/S]	PMAX? (Smax?)*
		P_{max}(S_{max})*	[R/S]	PMIN? (Smin?)*
		P_{min}(S_{min})*	[R/S]	FS?
		FS	[R/S]	
4	Calculate unknown			
	Syp		[XEQ] SYP	SYP=
	Se		[XEQ] SE	SE=
	Area		[XEQ] A	A=
	K		[XEQ] K	K=
	P_{max}		[XEQ] PMAX	PMAX=
	P_{min}		[XEQ] PMIN	PMIN=
	S_{max}		[XEQ] SMAX	SMAX=
	S_{min}		[XEQ] SMIN	SMIN=
	FS		[XEQ] FS	FS=
5	To change a value (then go to step 4)			
	Syp		[STO] 08	
	Se		[STO] 09	
	A		[STO] 01	
	K		[STO] 02	
	P_{max} (S_{max})		[STO] 03	
	P_{min} (S_{min})		[STO] 04	
	FS		[STO] 05	

*If S_{max} and S_{min} are to be input or calculated, use 1.00 for the value of area.

Program Listings

Initialization

```
01♦LBL "SEF"
02 ENG 3
03 "SYP?"
04 PROMPT
05 STO 08
06 "SE?"
07 PROMPT
08 STO 09
09 "A?"
10 PROMPT
11 STO 01
12 1
13 -
14 CF 00
15 X=0?
16 SF 00
17 "K?"
18 PROMPT
19 STO 02
20 "PMAX?"
21 FS? 00
22 "SMAX?"
23 PROMPT
24 STO 03
25 "PMIN?"
26 FS? 00
27 "SMIN?"
28 PROMPT
29 STO 04
30 "FS?"
31 PROMPT
32 STO 05
33 STOP
```

S_{yp}

```
34♦LBL "SYP"
35 XEQ 01
36 XEQ 02
37 RCL 02
38 *
39 RCL 09
40 /
41 CHS
42 RCL 05
43 1/X
44 +
45 /
46 STO 08
47 "SYP="
48 GTO 05
```

S_e

```
49♦LBL "SE"
50 XEQ 02
51 RCL 08
52 *
53 RCL 02
54 *
55 XEQ 01
56 CHS
57 RCL 08
58 RCL 05
59 /
60 +
61 /
62 STO 09
63 "SE="
64 GTO 05
```

S_{max}

P_{max}

```
65♦LBL "SMAX"
66♦LBL "PMAX"
67 RCL 01
68 ENTER↑
69 +
70 RCL 08
71 *
72 RCL 05
73 /
74 RCL 02
75 RCL 08
76 *
77 RCL 09
78 /
79 1
80 -
81 RCL 04
82 *
83 +
84 RCL 02
85 RCL 08
86 *
87 RCL 09
88 /
89 1
90 +
91 /
92 STO 03
93 "PMAX="
94 FS? 00
95 "SMAX="
96 GTO 05
```

Area

```
97♦LBL A
98 1
99 STO 01
100 XEQ 01
```

Program Listings

```
101 XEQ 02
102 RCL 02
103 *
104 RCL 08
105 *
106 RCL 09
107 /
108 +
109 RCL 08
110 /
111 RCL 05
112 *
113 STO 01
114 "A="
115 GTO 05
```

K

```
116♦LBL "K"
117 RCL 08
118 RCL 05
119 /
120 XEQ 01
121 -
122 XEQ 02
123 RCL 08
124 *
125 RCL 09
126 /
127 /
128 STO 02
129 "K="
130 GTO 05
```

Smin
Pmin

```
131♦LBL "SMIN"
132♦LBL "PMIN"
133 RCL 01
134 ENTER↑
135 +
136 RCL 08
137 *
138 RCL 05
139 /
140 RCL 02
141 RCL 08
142 *
143 RCL 09
144 /
145 1
146 +
147 RCL 03
148 *
149 -
150 1
151 RCL 02
152 RCL 08
153 *
154 RCL 09
155 /
156 -
157 /
158 STO 04
159 "PMIN="
160 FS? 00
161 "SMIN="
162 GTO 05
```

FS

```
163♦LBL "FS"
164 XEQ 01
165 XEQ 02
166 RCL 02
167 *
168 RCL 08
169 *
170 RCL 09
171 /
172 +
173 RCL 08
174 /
175 1/X
176 STO 05
177 "FS="
178 GTO 05
```

$Q=P_{max}-P_{min}$

```
179♦LBL 02
180 RCL 03
181 RCL 04
182 CHS
183 GTO 00
```

$Q=P_{max}-P_{min}$

```
184♦LBL 01
185 RCL 03
186 RCL 04
```

$\frac{Q}{2A}$

```
187♦LBL 00
188 +
189 RCL 01
190 /
191 2
192 /
193 RTN
```

Display

```
194♦LBL 05
195 ARCL X
196 AVIEW
197 STOP
198 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	A	50	
	K		
	P_{max} (S_{max})		
	P_{min} (S_{min})		
	FS		
05		55	
	S_{yp}		
	S_e		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 61 USER MODE
ENG 3 FIX ___ SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		S_{max}, S_{min}	P_{max}, P_{min}

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SPRING CONSTANT

This program calculates the value of any variable (X_1, F_1, X_2, F_2, k) given the other four in the spring equation. It may be used to solve any general linear equation of the form $y - y_o = m(x - x_o)$. It is also useful for linear interpolation in tables. Computed values are automatically stored to provide an interchangeable solution.

X_1 = Spring length

F_1 = Force required to retain spring at length X_1

X_2 = Spring length

F_2 = Force required to retain spring at length X_2

k = Spring constant

Equations:

$$k = \frac{F_1 - F_2}{X_2 - X_1}$$

$$F_1 = F_2 + k(X_2 - X_1)$$

$$F_2 = F_1 + k(X_1 - X_2)$$

$$X_1 = \frac{F_2 - F_1}{k} + X_2$$

$$X_2 = \frac{F_1 - F_2}{k} + X_1$$

Example 1:

A compression spring is 4.0 inches long under no compressive forces. A force of 270 lbf compresses the spring to a length of 2.8 inches. The solid height of the spring is 2.5 inches. Find the spring constant and the force required to fully compress the spring.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] SP [ALPHA]	X1?
4 [R/S]	F1?
0 [R/S]	X2?
2.8 [R/S]	F2?
270 [R/S]	K?
[R/S]	
[XEQ] [ALPHA] K [ALPHA]	K=225.00
2.5 [STO] 02	
[XEQ] [ALPHA] F2 [ALPHA]	F2=377.50

Example 2:

10.00%	10.25%	?
215.93	222.60	219.97

From the table shown, find the linear approximation to a value of 219.9749.

Keystrokes:	Display:
[XEQ] [ALPHA] SP [ALPHA]	X1?
10 [R/S]	F1?
215.93 [R/S]	X2?
10.25 [R/S]	F2?
222.60 [R/S]	K?
[R/S]	
[XEQ] [ALPHA] K [ALPHA]	K=-26.68
219.97 [STO] 03	
[XEQ] [ALPHA] X2 [ALPHA]	X2=10.15

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize		[XEQ] SP	X1?
3	Enter the unknown and skip the unknown	X1	[R/S]	F1?
	by pressing [R/S]	F1	[R/S]	X2?
		X2	[R/S]	F2?
		F2	[R/S]	K?
		k	[R/S]	
4	Calculate the unknown for:			
	X1		[XEQ] X1	X1 =
	F1		[XEQ] F1	F1 =
	X2		[XEQ] X2	X2 =
	F2		[XEQ] F2	F2 =
	K		[XEQ] K	K =
5	To vary one or more values:			
	X1	X1	[STO] 00	
	F1	F1	[STO] 01	
	X2	X2	[STO] 02	
	F2	F2	[STO] 03	
	K	k	[STO] 04	
	Repeat step 4			

Program Listings

```
01♦LBL "SP"
02 FIX 2          Initialization
03♦LBL A
04 "X1?"
05 PROMPT
06 STO 00
07♦LBL B
08 "F1?"
09 PROMPT
10 STO 01
11♦LBL C
12 "X2?"
13 PROMPT
14 STO 02
15♦LBL D
16 "F2?"
17 PROMPT
18 STO 03
19♦LBL E
20 "K?"
21 PROMPT
22 STO 04
23 STOP
24♦LBL "X1"       X1
25 XEQ 01
26 RCL 02
27 +
28 STO 00
29 "X1="
30 GTO 05
31♦LBL "F1"       F1
32 XEQ 00
33 RCL 03
34 +
35 STO 01
36 "F1="
37 GTO 05
38♦LBL "X2"       X2
39 XEQ 01
40 CHS
41 RCL 00
42 +
43 STO 02
44 "X2="
45 GTO 05
46♦LBL "F2"       F2
47 XEQ 00
48 CHS
49 RCL 01
50 +
51 STO 03
```

```
52 "F2="
53 GTO 05
54♦LBL "K"        k
55 RCL 01
56 RCL 03
57 -
58 RCL 02
59 RCL 00
60 -
61 /
62 STO 04
63 "K="           Display
64♦LBL 05
65 ARCL X
66 AVIEW
67 STOP
68♦LBL 01         (F2 - F1)/k
69 RCL 03
70 RCL 01
71 -
72 RCL 04
73 /
74 RTN
75♦LBL 00         k(X2 - X1)
76 RCL 02
77 RCL 00
78 -
79 RCL 04
80 *
81 RTN
82 .END.
```

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	X_1	50	
	F_1		
	X_2		
	F_2		
	k		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 27 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PROGRESSION OF A SLIDER CRANK

In a slider crank mechanism (e.g., the piston, wrist pin and connecting rod in an internal combustion engine), for given crank radius, connecting rod length, slider offset, crankshaft speed (RPM) and crank position, this program calculates the following: the displacement, velocity, and acceleration of the slider; the connecting rod angle, velocity and acceleration; the maximum and minimum displacements, and the maximum and minimum angular values for ϕ.



Equations:

$$\omega = \frac{\pi N}{30}$$

$$x = R \cos \theta + L \cos \phi$$

$$x_{max} = (R + L) \cos \left[\sin^{-1} \left(\frac{E}{R + L} \right) \right]$$

$$x_{min} = (L - R) \cos \left[\sin^{-1} \left(\frac{E}{L - R} \right) \right]$$

$$\Delta x = x_{max} - x_{min}$$

$$\phi = \sin^{-1} \left(\frac{E + R \sin \theta}{L} \right)$$

$$v = \frac{dx}{dt} = R\omega \left(\frac{-\sin (\theta + \phi)}{\cos \phi} \right)$$

$$a = \frac{d^2x}{dt^2} = R\omega^2 \left(\frac{-\cos\ (\theta + \phi)}{\cos\ \phi} - \frac{R\ \cos^2\ \theta}{L\ \cos^3\ \phi} \right)$$

$$\phi_{max} = \sin^{-1} \left(\frac{E + R}{L} \right)$$

$$\phi_{min} = \sin^{-1} \left(\frac{E - R}{L} \right)$$

$$\Delta\phi = \phi_{max} - \phi_{min}$$

$$\dot{\phi} = \frac{d\phi}{dt} = \omega\ \frac{R\ \cos\ \theta}{L\ \cos\ \phi}$$

$$\ddot{\phi} \quad \frac{d^2\phi}{dt^2} = \omega^2 \left[\left(\frac{d\phi}{d\theta} \right)^2 \tan\ \phi - \frac{R\ \sin\ \theta}{L\ \cos\ \phi} \right]$$

where:

N is crankshaft speed in RPM;

E is slider offset;

L is connecting rod length;

R is crank radius;

ω is crank angular velocity in radians/sec;

θ is crank angle;

x is slider displacement;

x_{max} is maximum slider displacement

x_{min} is minimum slider displacement;

Δx is stroke;

v is slider velocity;

a is slider acceleration;

ϕ is connecting rod angular displacement;

ϕ_{max} is maximum connecting rod angular displacement;

ϕ_{min} is minimum connecting rod angular displacement;

$\Delta\phi$ is total angular throw of connecting rod;

$\dot{\phi}$ is angular velocity of connecting rod;

$\ddot{\phi}$ is angular acceleration of connecting rod.

References:

H. A. Rothbart, Mechanical Design and Systems Handbook; McGraw-Hill, 1964.

V. M. Faires, Kinematics, McGraw-Hill; 1959.

Example 1:

For an in-line slider crank mechanism (E = 0), turning at 4800 RPM, having a crank radius of 2.0 inches and connecting rod length of 7.0 inches, find:

(1) x_{max}, x_{min} and ϕ_{max}, ϕ_{min}

(2) x, v, and a of the wrist pin in the slider

(3) $\dot{\phi}$, $\ddot{\phi}$, and ϕ of the connecting rod for θ = 0°, 15°, 45°, 90°, 135°, 180°,225°

The table below was produced by the following keystrokes:

θ°	x(in)	φ°	v(in/sec)	$\dot{\phi}$(rad/sec)	a(in/sec^2)	$\ddot{\phi}$(rad/sec^2)
0	9.00	0.00	0.00	143.62	-649701.96	0.00
15	8.91	4.24	-332.20	139.10	-614226.44	-17300.41
45	8.27	11.66	-857.50	103.69	-360454.40	-49902.29
90	6.71	16.60	-1005.31	0.00	150658.43	-75329.22
135	5.44	11.66	-564.22	-103.69	354181.29	-49902.29
180	5.00	0.00	0.00	-143.62	360945.53	0.00
225	5.44	-11.66	564.22	-103.69	354181.29	49902.29

Keystrokes:	Display:	
[USER]		(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 016		
[XEQ] [ALPHA] PSC [ALPHA]	N?	
4800 [R/S]	E?	
0 [R/S]	L?	
7 [R/S]	R?	
2 [R/S]	W=502.65	
[A]	X=9.00	(X_{max})
[R/S]	X=5.00	(X_{min})
[B]	∡=16.60	(max)
[R/S]	∡=-16.60	(min)
[C]	∡?	
0 [R/S]	X=9.00	

Keystrokes:	Display:
[R/S]	∡=0.00
[D]	V=0.00
[R/S]	d∡=143.62
[E]	A=-649,701.96
[R/S]	d↑2∡=0.00
[C]	∡?
15 [R/S]	X=8.91
[R/S]	∡=4.24
[D]	V=-332.20
[R/S]	d∡=139.10
⋮	⋮
[C]	∡?
225 [R/S]	X=5.44
[R/S]	∡=-11.66
[D]	V=564.22
[R/S]	d∡=-103.69
[E]	A=354,181.29
[R/S]	d↑2∡=49,902.29

User Instructions

				SIZE: 016
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution and input:		[XEQ] PSC	N?
	Key in crankshaft speed in RPM	N	[R/S]	E?
	Key in slider offset	E	[R/S]	L?
	Key in connecting rod length	L	[R/S]	R?
	Key in crank radius	R	[R/S]	W=(ω)
3	For X_{max}, X_{min}		[A]	X=(X_{max})
			[R/S]	X=(X_{min})
4	For ϕ_{max}, ϕ_{min}		[B]	∡=(θ_{max})
			[R/S]	∡=(θ_{min})
5	For X, ϕ, V, $\dot{\phi}$, a, $\ddot{\phi}$:		[C]	∡?
	Key in θ	θ	[R/S]	X=
			[R/S]	∡=(ϕ)
	then		[D]	V=
			[R/S]	d∡=($\dot{\phi}$)
	then		[E]	A=
			[R/S]	d↑2∡=($\ddot{\phi}$)
6	For automatic display:		[F]	∡2?
	Key in ending θ	θ_2	[R/S]	∡1?
	Key in beginning θ	θ_1	[R/S]	N?
	Key in the number of sectors you wish	n	[R/S]	∡=(θ)
	to divide the interval into		[R/S]	X=
			[R/S]	∡=(ϕ)
			[R/S]	V=
			[R/S]	d∡=($\dot{\phi}$)
			[R/S]	A=
	(continue pressing [R/S] until done.)		[R/S]	∡d↑2=($\ddot{\phi}$)

Program Listings

```
LBL "PSC"
```

Initialization

```
FIX 2
"N?"
PROMPT
STO 11
"E?"
PROMPT
STO 12
"L?"
PROMPT
STO 13
"R?"
PROMPT
STO 14
```

Calculate ω

```
RCL 11
PI
*
30
/
STO 15
"W="
GTO 05
```

Xmax + Xmin

```
LBL A
CF 01
XEQ 00
SF 01
STO 04
XEQ 00
STOP
```

Ømax + Ømin

```
LBL B
1
ASIN
XEQ 02
STO 04
"∡="
XEQ 05
-1
ASIN
XEQ 02
"∡="
GTO 05
```

Calculate X + Ø

```
LBL C
"∡?"
PROMPT
STO 00
LBL 07
XEQ 02
COS
STO 04
RCL 13
*
RCL 00
COS
STO 03
RCL 14
*
+
STO 01
"X="
XEQ 05
RCL 00
XEQ 02
"∡="
GTO 05
```

Calculate V + Ø

```
LBL D
RCL 00
RCL 02
+
SIN
CHS
RCL 04
/
RCL 15
*
RCL 14
*
"V="
XEQ 05
RCL 04
RCL 13
*
RCL 14
/
1/X
STO 05
RCL 03
*
RCL 15
*
"d∡="
GTO 05
```

Calculate A + Ø

```
LBL E
RCL 03
X↑2
RCL 14
*
RCL 13
/
RCL 04
3
Y↑X
/
```

Program Listings

```
103 RCL 00
104 RCL 02
105 +
106 COS
107 RCL 04
108 /
109 +
110 CHS
111 RCL 14
112 *
113 RCL 15
114 X↑2
115 *
116 "A="
117 XEQ 05
118 RCL 02
119 TAN
120 RCL 05
121 RCL 03
122 *
123 X↑2
124 *
125 RCL 00
126 SIN
127 RCL 05
128 *
129 -
130 RCL 15
131 X↑2
132 *
133 "d↑2∡="
134 GTO 05
```

Automatic looping

```
135♦LBL F
136 "∡2?"
137 PROMPT
138 STO 08
139 "∡1?"
140 PROMPT
141 STO 07
142 "N?"
143 PROMPT
144 STO 09
145 STO 10
146 ISG 10
147 RDN
148 RCL 07
149 RCL 08
150 -
151 CHS
152 RCL 09
153 /
154 STO 06
155♦LBL 04
156 RCL 07
157 "∡="
158 ARCL X
159 AVIEW
160 STOP
161 STO 00
162 CF 01
163 XEQ 07
164 XEQ D
165 XEQ E
166 SF 01
167 DSE 10
168 GTO 03
169 RTN
170♦LBL 03
171 RCL 06
172 ST+ 07
173 GTO 04
174 RTN
```

display

```
175♦LBL 05
176 ARCL X
177 AVIEW
178 STOP
179 RTN
```

Calculate Xmax + Xmin

```
180♦LBL 00
181 RCL 12
182 RCL 13
183 RCL 14
184 FS? 01
185 CHS
186 +
187 /
188 ASIN
189 COS
190 RCL 13
191 RCL 14
192 FS? 01
193 CHS
194 +
195 *
196 "X="
197 XEQ 05
198 RTN
199♦LBL 02
200 SIN
201 RCL 14
202 *
203 RCL 12
204 +
```

Program Listings

```
205 RCL 13
206 /
207 ASIN
208 STO 02
209 RTN
210 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	θ	50	
	x		
	φ		
	COS θ		
	COS φ		
05	R/(L COS φ)	55	
	$(\theta_1-\theta_2)/h$		
	θ_1		
	θ_2		
	n		
10	Index	60	
	N		
	E		
	L		
	R		
15	ω	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 61 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		Maximum	Minimum

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

FREE VIBRATIONS

This program provides an exact solution to the differential equation for a damped oscillator vibrating freely: $m\ddot{x} + c\dot{x} + kx = 0$.

The user inputs the mass m, spring constant k, and damping constant c. The output will be:

1. ω for an underdamped system, i.e. $c < c_{crit}$ · c_{crit} is calculated by pressing [B].

2. 0 for a critically damped system, i.e. $c = c_{crit}$.

3. -1 for an overdamped system, i.e. $c > c_{crit}$.

The initial conditions are the displacement and velocity at time zero (x_o and $\dot{x}_o$).

Equations:

$$c_{crit} = 2\sqrt{km}$$

$$\omega = \sqrt{\frac{k}{m} - \left(\frac{c}{2m}\right)^2}$$

$$\ddot{x} = -(c\dot{x} + kx)/m$$

Underdamping $\qquad (c^2 - 4km < 0)$

$$x(t) = Re^{-\frac{c}{2m}t} \cos(\omega t - \delta)$$

$$\dot{x}(t) = -R\omega e^{-\frac{c}{2m}t} \sin(\omega t - \delta) - \frac{c}{2m} Re^{-\frac{c}{2m}t} \cos(\omega t - \delta)$$

where:

$$R \cos \delta = x_o$$

$$R \sin \delta = \frac{1}{\omega}\left[\dot{x}_o + \frac{c}{2m}\, x_o\right]$$

Critical damping

$$(c = c_{crit},\ \text{or}\ c^2 = 4km)$$

$$x(t) = (A_{cr} + B_{cr}t)e^{-\frac{c}{2m}t}$$

$$\dot{x}(t) = \left[\, B_{cr} - \frac{c}{2m}(A_{cr} + B_{cr}t)\right] e^{-\frac{c}{2m}t}$$

where:

$$A_{cr} = x_o$$

$$B_{cr} - \dot{x}_o + \frac{c}{2m}\, x_o$$

Overdamping

$$(c^2 - 4km > 0)$$

$$\dot{x}(t) = A_{ov}e^{r_1 t} + B_{ov}e^{r_2 t}$$

$$x(t) = A_{ov}r_1 e^{r_1 t} + B_{ov}r_2 e^{r_2 t}$$

where:

$$r_1,\ r_2 = -\frac{c}{2m} \pm \sqrt{\left(\frac{c}{2m}\right)^2 - \frac{k}{m}}$$

$$A_{ov} = x_o - B_{ov}$$

$$B_{ov} = \frac{\dot{x}_o - r_1 x_o}{r_2 - r_1}$$



Reference:

Boyce, W. E. and DiPrima, R. C., Elementary Differential Equations; John Wiley and Sons, 1969.

Remarks:

For overdamping, ω has no meaning and is, in fact, an imaginary number.

For $c = c_{crit}$, $\omega = 0$.

This program sets the angular mode of the calculator to radians. Erroneous answers will occur if the degrees mode is inadvertently set.

Example:

A mass of 20 g stretches a spiral spring 10 cm. The mass is pulled down an additional 4 cm, held, and then released. Find the mass displacement and velocity at 0.1 second intervals up to 1 second for c = 50 dyne-sec/cm.

$$k = \frac{F}{x} = \frac{mg}{x} = \frac{20g\ (980\ cm/s^2)}{10\ cm} = \frac{20 \times 980}{10}\ dyne/cm$$

Keystrokes:	Display:
[USER]	
[XEQ] [ALPHA] SIZE [ALPHA] 016	
[XEQ] [ALPHA] VIB [ALPHA]	M?
20 [R/S]	C?
50 [R/S]	K?
20 [ENTER↑] 980 [x] 10 [÷] [R/S]	W=9.820
[B]	C=395.980
[C]	X?
4 [R/S]	dX?
0 [R/S]	T?
0 [R/S]	X=4.000
[R/S]	dX=-1.000E-9
[R/S]	d2X=-392.000
[D]	T?
0.1 [R/S]	X=2.334
[R/S]	dX=-29.296
[R/S]	d2X=-155.493
[D]	T?
⋮	⋮
(Repeat for t = .2, .3 ... 1.0)	
⋮	⋮
1 [R/S]	X=-1.114
[R/S]	dX=4.406
[R/S]	d2X=98.132

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution		[XEQ] VIB	M?
		m	[R/S]	C?
		c	[R/S]	K?
	underdamped system	k	[R/S]	W=
	critically damped			0.000
	overdamped system			-1.000
3	To calculate C_{crit}		[B]	C=
4	Input initial conditions $x(0)$ and $\dot{x}(0)$		[C]	X?
		X(0)	[R/S]	dX?
		$\dot{X}(0)$	[R/S]	T?
5	Input t and calculate $x(t)$, $\dot{x}(t)$ and $\ddot{x}(t)$	t	[R/S]	X=
			[R/S]	dX=
			[R/S]	d2X=
6	For each new t		[D]	T?
	Go to step 5			

Program Listings

Initialization

```
01♦LBL "VIB"
02 RAD
03 FIX 3
04 "M?"
05 PROMPT
06 STO 02
07 "C?"
08 PROMPT
09 STO 03
10 "K?"
11 PROMPT
```

Calculate

$$\left[\frac{k}{m} - \left(\frac{c}{2m}\right)^2\right]^{1/2}$$

```
12 STO 04
13 RCL 02
14 /
15 STO 15
16 RCL 03
17 RCL 02
18 2
19 *
20 /
21 STO 14
22 X↑2
23 -
24 RND
25 X<0?
26 GTO 01
27 X=0?
28 STOP
29 SQRT
30 STO 05
31 "W="
32 GTO 05
33♦LBL 01
34 STO 05
35 -1
36 STOP
```

Calculate C_{cr}

```
37♦LBL B
38 RCL 04
39 RCL 02
40 *
41 SQRT
42 2
43 *
44 "C="
45 GTO 05
```

Initial conditions

```
46♦LBL C
47 "X?"
48 PROMPT
49 STO 00
50 "dX?"
51 PROMPT
52 STO 01
```

Calculate x(t), $\dot{x}(t)$ and $\ddot{x}(t)$

```
53♦LBL D
54 "T?"
55 PROMPT
56 STO 07
57 RCL 14
58 *
59 CHS
60 E↑X
61 STO 13
62 RCL 05
63 X<0?
64 GTO c
65 X=0?
66 GTO b
```

$C > C_{cr}$

```
67♦LBL a
68 RCL 14
69 RCL 00
70 *
71 RCL 01
72 +
73 RCL 05
74 /
75 RCL 00
76 R-P
77 STO 11
78 RDN
79 STO 12
80 RCL 05
81 RCL 07
82 *
83 -
84 CHS
85 STO 09
86 COS
87 RCL 13
88 *
89 RCL 11
90 *
91 STO 02
92 "X="
93 XEQ 05
94 RCL 14
95 *
96 RCL 09
97 SIN
98 RCL 13
99 *
100 RCL 05
101 *
102 RCL 11
```

Program Listings

```
*
+
CHS
"dX="
XEQ 05
GTO e
```

$C = C_{cr}$

```
◆LBL b
RCL 14
RCL 00
*
RCL 01
+
STO 12
RCL 07
*
RCL 00
+
RCL 13
*
STO 02
"X="
XEQ 05
RCL 14
*
CHS
RCL 12
RCL 13
*
+
"dX="
XEQ 05
GTO e
```

$C < C_{cr}$

```
◆LBL c
CHS
SQRT
ENTER↑
ENTER↑
RCL 14
-
STO 03
X<>Y
2
*
-
STO 04
RCL 01
RCL 03
RCL 00
*
-
RCL 04
RCL 03
-
/
STO 12
RCL 00
-
CHS
STO 11
RCL 03
RCL 07
*
E↑X
RCL 11
*
STO 02
RCL 03
*
STO 08
RCL 04
RCL 07
*
E↑X
RCL 12
*
ST+ 02
RCL 04
*
ST+ 08
RCL 02
"X="
XEQ 05
RCL 08
"dX="
XEQ 05
```

Calculate acceleration

```
◆LBL e
RCL 14
2
*
*
RCL 02
RCL 15
*
+
CHS
"d2X="
```

Display

```
◆LBL 05
ARCL X
AVIEW
STOP
RTN
.END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x_0	50	
	$\dot{x}_0$		
	m, x(t)		
	c, r_1		
	k, r_2		
05	ω, $\sqrt{(-\omega)^2}$	55	
	$(t_1-t_2)/n$		
	t, t_1		
	$\dot{x}(t)$		
	$\omega t-\delta$		
10	n	60	
	R, A_{cr}, A_{ov}		
	δ, B_{cr}, B_{ov}		
	$e^{-c/2m\,t}$		
	c/2m		
15	k/m	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 56 USER MODE

ENG ___ FIX 3 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INTERFERENCE FITS

This program can be used to determine contact pressure or interference for concentric cylinders. Once the contact pressure has been determined, the actual tangential stresses at the surfaces of the cylinders can be determined. These stresses may be used in maximum shear theory of failure analysis. Modified tangential stresses for use in maximum strain theory of failure analysis can also be computed.

Concentric Cylinders



Equations:

for contact pressure:

$$\delta = d_c P_c \left[\frac{d_c^{\,2} + d_i^{\,2}}{E_i(d_c^{\,2} - d_i^{\,2})} + \frac{d_o^{\,2} + d_c^{\,2}}{E_o(d_o^{\,2} - d_c^{\,2})} - \frac{\mu_i}{E_i} + \frac{\mu_o}{E_o} \right]$$

for actual tangential stresses:

$$S_{to} = \frac{2P_c d_c^{\,2}}{d_o^{\,2} - d_c^{\,2}}$$

$$S_{tco} = P_c \left(\frac{d_o^{\,2} + d_c^{\,2}}{d_o^{\,2} - d_c^{\,2}} \right)$$

$$S_{tci} = -P_c \left(\frac{d_c^{\,2} + d_i^{\,2}}{d_c^{\,2} - d_i^{\,2}} \right)$$

$$S_{ti} = \frac{-2P_c d_c^2}{d_c^2 - d_i^2}$$

for modified tangential stresses:

$$S'_{to} = \frac{2P_c d_c^2}{d_o^2 - d_c^2}$$

$$S'_{tco} = P_c \left(\frac{d_o^2 + d_c^2}{d_o^2 - d_c^2} + \mu_o \right)$$

$$S'_{tci} = -P_c \left(\frac{d_c^2 + d_i^2}{d_c^2 - d_i^2} - \mu_i \right)$$

$$S'_{ti} = \frac{-2P_c d_c^2}{d_c^2 - d_i^2}.$$

where:

δ is the total interference;

P_c is the contact pressure;

d_i is the inside diameter;

d_c is the contact surface diameter;

d_o is the outside diameter;

μ_o is Poisson's ratio for the outside material;

μ_i is Poisson's ratio for the inside material;

E_o is the modulus of elasticity for the outside material;

E_i is the modulus of elasticity for the inside material;

S_{to} is the tangential stress of the outer surface;

S_{tco} is the tangential stress in the outer material at the contact surface;

S_{tci} is the tangential stress in the inner material at the contact surface;

S_{ti} is the stress at the inner surface of the inner cylinder;

S'_{to}, S'_{tco}, S'_{tci}, and S'_{ti} are the modified tangential stresses corresponding to the actual stresses just described.

Reference:

Hall, Holowenko, Laughlin,
Theory and Problems of Machine Design; Schaum's Outline Series, McGraw Hill Co., 1961.

Example 1:

The choke at the end of a shotgun barrel is to be attached using an interference fit. If 5000 pounds per square inch must be developed to hold the choke in place, what is the minimum allowable interference? What are the values or actual stress?

$d_i = 0.75$ in

$d_c = 0.9375$ in

$d_o = 1.125$ in

$\mu_o = \mu_i = 0.3$

$E_o = E_i = 30 \times 10^6$ psi

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] INTER [ALPHA]	DI?
0.75 [R/S]	DC?
0.9375 [R/S]	DO?
1.125 [R/S]	RATIO O?
0.3 [R/S]	EO?

Keystrokes:	Display:
30 [EEX] 6 [R/S]	RATIO I?
0.3 [R/S]	EI?
30 [EEX] 6 [R/S]	Pc?
5000 [R/S]	DELTA=1.578E-3
[R/S]	STO=22.73E3
[R/S]	STCO=27.73E3
[R/S]	STCI=-22.78E3
[R/S]	STI=-27.78E3

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution			
	to find interference		[XEQ] INTER	DI?
	to find contact pressure		[XEQ] PRES	DI?
3	Input data			
		d_i	[R/S]	DC?
		d_c	[R/S]	DO?
		d_o	[R/S]	RATIO O?
		μ_o	[R/S]	EO?
		E_o	[R/S]	RATIO I?
		μ_i	[R/S]	EI?
		E_i	[R/S]	DELTA? or Pc?
		δ or P_c	[R/S]	
4	Output is automatic			Pc=
				(or)
				DELTA=
5	Continue to find:			
	S_{to}		[R/S]	STO=
	S_{tco}		[R/S]	STCO=
	S_{tci}		[R/S]	STCI=
	S_{ti}		[R/S]	STI=
	S'_{to}		[R/S]	S:TO=
	S'_{tco}		[R/S]	S:TCO=
	S'_{tci}		[R/S]	S:TCI=
	S'_{ti}		[R/S]	S:TI=

Program Listings

Initialization

```
 01♦LBL "INT
ER"
 02 SF 00
 03 GTO 00
 04♦LBL "PRE
S"
 05 CF 00
 06♦LBL 00
 07 ENG 3
 08 "DI?"
 09 PROMPT
 10 STO 01
 11 "DC?"
 12 PROMPT
 13 STO 02
 14 "DO?"
 15 PROMPT
 16 STO 03
 17 "RATIO O
?"
 18 PROMPT
 19 STO 04
 20 "EO?"
 21 PROMPT
 22 STO 05
 23 "RATIO I
?"
 24 PROMPT
 25 STO 06
 26 "EI?"
 27 PROMPT
 28 STO 07
 29 FS? 00
 30 GTO A
 31 "DELTA?"
 32 PROMPT
 33 STO 08
 34 XEQ E
 35 RCL 08
 36 X<>Y
 37 /
 38 STO 08
 39 "Pc="
 40 XEQ d
 41 GTO B
 42♦LBL A
 43 "Pc?"
 44 PROMPT
 45 STO 08
 46 XEQ E
 47 RCL 08
 48 *
 49 "DELTA="
 50 XEQ d
 51 GTO B
```

Subroutine "E"

```
 52♦LBL E
 53 RCL 01
 54 X↑2
 55 RCL 02
 56 X↑2
 57 +
 58 LASTX
 59 RCL 01
 60 X↑2
 61 -
 62 /
 63 RCL 07
 64 /
 65 RCL 02
 66 X↑2
 67 RCL 03
 68 X↑2
 69 +
 70 LASTX
 71 RCL 02
 72 X↑2
 73 -
 74 /
 75 RCL 05
 76 /
 77 +
 78 RCL 06
 79 RCL 07
 80 /
 81 -
 82 RCL 04
 83 RCL 05
 84 /
 85 +
 86 RCL 02
 87 *
 88 RTN
```

Outputs

```
 89♦LBL B
 90 0
 91 RCL 03
 92 XEQ E
 93 RCL 08
 94 *
 95 2
 96 *
 97 "STO="
 98 XEQ d
 99 RCL 03
100 RCL 03
```

Program Listings

```
101 XEQ E
102 RCL 08
103 *
104 "STCO="
105 XEQ d
106 RCL 01
107 RCL 01
108 XEQ E
109 RCL 08
110 *
111 "STCI="
112 XEQ d
113 0
114 RCL 01
115 XEQ E
116 RCL 08
117 *
118 2
119 *
120 "STI="
121 XEQ d
122 0
123 RCL 03
124 XEQ E
125 RCL 08
126 *
127 2
128 *
129 "S:TO="
130 XEQ d
131 RCL 03
132 RCL 03
133 XEQ E
134 RCL 04
135 +
136 RCL 08
137 *
138 "S:TCO="
139 XEQ d
140 RCL 01
141 RCL 01
142 XEQ E
143 RCL 06
144 +
145 RCL 08
146 *
147 "S:TCI="
148 XEQ d
149 0
150 RCL 01
151 XEQ E
152 RCL 08
```

```
153 *
154 2
155 *
156 "S:TI="
157 GTO d
```

Subroutine "E"

```
158◆LBL E
159 X↑2
160 X<>Y
161 X↑2
162 RCL 02
163 X↑2
164 +
165 X<>Y
166 LASTX
167 -
168 /
169 RTN
```

Display

```
170◆LBL d
171 ARCL X
172 PROMPT
173 RTN
174 END
```

80	
90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	d_i		
	d_c		
	d_0		
	μ_0		
05	E_0	55	
	μ_i		
	E_i		
	P_c, δ		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 57 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Calculate δ	Calculate P_c

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

LINEAR OR ANGULAR DEFORMATION

This program solves for linear deflection under tensile load, or the analogous angular deflection under torque, using an interchangeable solution. Given four of the five variables, the unknown is calculated.

Equations:



$$\Delta\ell = \frac{P\ell}{AE} \qquad \theta = \frac{T\ell}{JG}$$

where:

$\Delta\ell$ is the change in length;

P is the applied load;

ℓ is the length;

A is the cross sectional area;

E is the modulus of elasticity;

θ is the deflection angle in radians;

T is the applied torque;

J is the polar moment of the section;

G is the modulus of elasticity in shear.

Remarks:

This program is not applicable for non-elastic media or elastic media where stress exceeds the elastic limit. Materials must be isotropic. The equation for angular deflection is not valid in the neighborhood of the applied torque.

Example:

Steel bars affixed to the roof are to be used to support the end of a cantilever balcony. The load on each bar will be 50,000 newtons. If the maximum allowable deflection is 0.001 meters, what should the area of the bars be? ℓ = 10 meters E = 2.068 X 10" N/m^2

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] LINEAR [ALPHA]	A
[R/S]	DELTA
.001 [R/S]	L
10 [R/S]	P
50,000 [R/S]	E
2.068 [EEX] 11 [R/S]	A=2.418E-3

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input data (skip			
	unknown by pressing [R/S])			
	for linear system		[XEQ] LINEAR	A
	Key in area	A	[R/S]	DELTA
	Key in change in length	$\Delta \ell$	[R/S]	L
	Key in initial length	ℓ	[R/S]	P
	Key in applied load	P	[R/S]	E
	Key in modulus of elasticity	E	[R/S]	(unknown)=
	or for angular system		[XEQ] ANGULAR	J
	Key in polar moment of the section	J	[R/S]	$\angle$
	Key in the deflection angle	θ (rads)	[R/S]	L
	Key in the length	ℓ	[R/S]	T
	Key in the applied torque	T	[R/S]	G
	Key in the modulus of elasticity	G	[R/S]	(unknown)=
	in shear			

Program Listings

Initialization "LINEAR"

```
01♦LBL "LIN
EAR"
02 CF 00
03 CLX
04 "A"
05 PROMPT
06 X=0?
07 XEQ b
08 STO 01
09 CLX
10 "DELTA"
11 PROMPT
12 X=0?
13 XEQ b
14 STO 02
15 CLX
16 "L"
17 PROMPT
18 X=0?
19 XEQ a
20 STO 03
21 CLX
22 "P"
23 PROMPT
24 X=0?
25 XEQ a
26 STO 04
27 CLX
28 "E"
29 PROMPT
30 X=0?
31 XEQ b
32 STO 05
33 GTO 00
```

Initialization "ANGULAR"

```
34♦LBL "ANG
ULAR"
35 CF 00
36 CLX
37 "J"
38 PROMPT
39 X=0?
40 XEQ b
41 STO 01
42 CLX
43 "∡"
44 PROMPT
45 X=0?
46 XEQ b
47 STO 02
48 CLX
49 "L"
50 PROMPT
51 X=0?
52 XEQ a
53 STO 03
54 CLX
55 "T"
56 PROMPT
57 X=0?
58 XEQ a
59 STO 04
60 CLX
61 "G"
62 PROMPT
63 X=0?
64 XEQ b
65 STO 05
66 GTO 00
```

Store output label

```
67♦LBL a
68 SF 00
69♦LBL b
70 ASTO 00
71 1
72 RTN
```

Computation

```
73♦LBL 00
74 ENG 3
75 RCL 04
76 RCL 03
77 *
78 RCL 02
79 RCL 01
80 *
81 RCL 05
82 *
83 /
84 FS? 00
85 1/X
```

Display

```
86 CLA
87 ARCL 00
88 "⊢="
89 ARCL X
90 AVIEW
91 STOP
92 .END.
```

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	OUTPUT LABEL	50	
	A(J)		
	DELTA (Δ)		
	L		
	P(T)		
05	E(G)	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 28 USER MODE
ENG 3 FIX ____ SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Invert	Don't Invert

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONSTANT ACCELERATION

This program calculates an interchangeable solution among the variables displacement, acceleration, initial velocity, and time, for an object that undergoes constant acceleration. The motion may be either circular or linear. Final velocity as a function of initial velocity, acceleration, and displacement may also be computed.

Equations:

	Linear	Angular
Displacement	$x = v_o t + \frac{1}{2} at^2$	$\theta = \omega_o t + \frac{1}{2} \alpha t^2$
Initial velocity	$v_o = \frac{x}{t} - \frac{1}{2} at$	$\omega_o = \frac{\theta}{t} - \frac{1}{2} \alpha t$
Acceleration	$a = \frac{x - v_o t}{\frac{1}{2} t^2}$	$\alpha = \frac{\theta - \omega_o t}{\frac{1}{2} t^2}$
Time	$t = \frac{\sqrt{v_o^2 + 2ax} - v_o}{a}$	$t = \frac{\sqrt{\omega_o^2 + 2\alpha\theta} - \omega_o}{\alpha}$
Final velocity	$v = \sqrt{v_o^2 + 2ax}$	$\omega = \sqrt{\omega_o^2 + 2\alpha\theta}$

Remarks:

Any consistent set of units may be used.

Displacement, acceleration, and velocity should be considered as signed (vector) quantities. For example, if initial velocity and acceleration are in opposite directions, one should be positive and the other negative.

All equations assume that the initial displacement, x_o or θ_o, equals zero.

Example 1:

An automobile accelerates for 4 seconds from a speed of 35 mph and covers a distance of 264 feet. Assuming constant acceleration, what is the acceleration in ft/sec^2? If the acceleration continues to be constant, what distance is covered in the next second?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] CA [ALPHA]	X,∠?
264 [R/S]	VO,WO?
35 [ENTER] 5280 [*] 3600 [÷] [R/S]	T?
4 [R/S]	A?
[R/S]	
[XEQ] A	A=7.733
5 [STO] 02	
[XEQ] X	X,∠=348.33 (total distance)
264 [-]	84.33

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution		[XEQ] CA	X,∠?
	(input 3 of the 4 items skipping the	X or θ	[R/S]	VO,WO?
	unknown by pressing [R/S])	V or ω	[R/S]	T?
		t	[R/S]	A?
		A or α	[R/S]	
3	Calculate the unknown			
	Displacement		[XEQ] X	X,∠=
	Initial velocity		[XEQ] VO	VO,WO=
	Acceleration		[XEQ] A	A=
	Time		[XEQ] T	T=
	Final velocity		[XEQ] V	V,W=
4	To change any value(s)	X or θ	[STO] 00	
		V or ω	[STO] 01	
		t	[STO] 02	
		A or α	[STO] 03	
5	Go to step 3			

Program Listings

Initialization

```
♦LBL "CA"
FIX 2
"X,≮?"
PROMPT
STO 00
"VO,WO?"
PROMPT
STO 01
"T?"
PROMPT
STO 02
"A?"
PROMPT
STO 03
STOP
```

Displacement

```
♦LBL "X"
RCL 02
X↑2
RCL 03
*
2
/
RCL 02
RCL 01
*
+
STO 00
"X,≮="
GTO 05
```

Initial velocity

```
♦LBL "VO"
RCL 00
RCL 02
/
RCL 02
RCL 03
*
2
/
-
STO 01
"VO,WO="
GTO 05
```

Acceleration

```
♦LBL A
RCL 00
RCL 01
RCL 02
*
-
RCL 02
X↑2
2
/
/
STO 03
"A="
GTO 05
```

Time

```
♦LBL "T"
CF 01
XEQ 01
RCL 01
-
RCL 03
/
STO 02
"T="
GTO 05
```

Final velocity

```
♦LBL "V"
SF 01
♦LBL 01
RCL 01
X↑2
RCL 03
RCL 00
*
2
*
+
SQRT
FC? 01
RTN
"V,W="
```

Display

```
♦LBL 05
ARCL X
AVIEW
STOP
.END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	X, θ	50	
	V or ω		
	t		
	A, α		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 25 USER MODE
ENG ___ FIX 2 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Mechanical Engineering

MECHANICAL ENGINEERING

GEAR FORCES..1

STRESS ON AN ELEMENT...2

EQUATIONS OF STATE...4

SODERBERG'S EQUATION FOR FATIGUE...............................6

SPRING CONSTANT..8

PROGRESSION OF A SLIDER CRANK..................................9

FREE VIBRATIONS...11

INTERFERENCE FITS...13

LINEAR OR ANGULAR DEFORMATION.................................15

CONSTANT ACCELERATION...16

GEAR FORCES

PROGRAM REGISTERS NEEDED: 31

ROW 1 (1 - 3)

ROW 2 (4 - 10)

ROW 3 (11 - 14)

ROW 4 (15 - 21)

ROW 5 (22 - 31)

ROW 6 (31 - 35)

ROW 7 (36 - 37)

ROW 8 (37 - 49)

ROW 9 (50 - 61)

ROW 10 (61 - 64)

ROW 11 (64 - 66)

ROW 12 (67 - 77)

ROW 13 (78 - 86)

ROW 14 (86 - 97)

ROW 15 (98 - 106)

ROW 16 (106 - 113)

ROW 17 (114 - 119)

STRESS ON AN ELEMENT

PROGRAM REGISTERS NEEDED: 46

ROW 1 (1 - 2)

ROW 2 (3 - 6)

ROW 3 (7 - 15)

ROW 4 (15 - 20)

ROW 5 (21 - 27)

ROW 6 (27 - 36)

ROW 7 (37 - 48)

ROW 8 (49 - 60)

ROW 9 (61 - 68)

ROW 10 (68 - 78)

ROW 11 (79 - 89)

ROW 12 (90 - 101)

ROW 13 (102 - 110)

ROW 14 (111 - 115)

ROW 15 (115 - 120)

ROW 16 (120 - 128)

ROW 17 (129 - 139)

ROW 18 (140 - 148)

ROW 19 (149 - 155)

ROW 20 (156 - 162)

ROW 21 (163 - 169)

ROW 22 (170 - 180)

ROW 23 (180 - 189)

ROW 24 (190 - 200)

ROW 25 (200 - 204)

EQUATIONS OF STATE

PROGRAM REGISTERS NEEDED: 47

ROW 1 (1 - 5)

ROW 2 (5 - 10)

ROW 3 (11 - 17)

ROW 4 (18 - 24)

ROW 5 (25 - 30)

ROW 6 (30 - 37)

ROW 7 (38 - 47)

ROW 8 (48 - 55)

ROW 9 (55 - 65)

ROW 10 (66 - 73)

ROW 11 (73 - 79)

ROW 12 (80 - 91)

ROW 13 (91 - 98)

ROW 14 (98 - 108)

ROW 15 (109 - 121)

ROW 16 (122 - 130)

ROW 17 (130 - 138)

ROW 18 (139 - 151)

ROW 19 (152 - 164)

ROW 20 (165 - 177)

ROW 21 (178 - 190)

ROW 22 (191 - 202)

ROW 23 (203 - 210)

ROW 24 (211 - 222)

ROW 25 (222 - 230)

ROW 26 (230 - 230)

SODERBERG'S EQUATION FOR
FATIGUE
PROGRAM REGISTERS NEEDED: 52

ROW 1 (1 – 3)

ROW 2 (3 – 10)

ROW 3 (11 – 19)

ROW 4 (20 – 22)

ROW 5 (22 – 27)

ROW 6 (27 – 33)

ROW 7 (34 – 36)

ROW 8 (37 – 47)

ROW 9 (47 – 50)

ROW 10 (51 – 61)

ROW 11 (62 – 65)

ROW 12 (65 – 69)

ROW 13 (70 – 82)

RCW 14 (83 – 93)

ROW 15 (93 – 96)

ROW 16 (97 – 104)

ROW 17 (105 – 115)

ROW 18 (115 – 121)

SODERBERG'S EQUATION FOR FATIGUE

ROW 19 (122 - 130)

ROW 20 (130 - 132)

ROW 21 (132 - 141)

ROW 22 (142 - 154)

ROW 23 (155 - 161)

ROW 24 (161 - 163)

ROW 25 (164 - 172)

ROW 26 (173 - 181)

ROW 27 (182 - 193)

ROW 28 (194 - 198)

SPRING CONSTANT

PROGRAM REGISTERS NEEDED: 23

ROW 1 (1 - 4)

ROW 2 (4 - 11)

ROW 3 (12 - 17)

ROW 4 (18 - 24)

ROW 5 (24 - 30)

ROW 6 (30 - 35)

ROW 7 (36 - 39)

ROW 8 (39 - 46)

ROW 9 (46 - 52)

ROW 10 (52 - 57)

ROW 11 (58 - 67)

ROW 12 (68 - 80)

ROW 13 (81 - 82)

PROGRESSION OF A SLIDER CRANK

PROGRAM REGISTERS NEEDED: 46

ROW 1 (1 - 4)

ROW 2 (5 - 12)

ROW 3 (12 - 21)

ROW 4 (22 - 28)

ROW 5 (28 - 35)

ROW 6 (35 - 40)

ROW 7 (41 - 47)

ROW 8 (48 - 59)

ROW 9 (59 - 64)

ROW 10 (65 - 76)

ROW 11 (77 - 85)

ROW 12 (86 - 93)

ROW 13 (94 - 106)

ROW 14 (107 - 117)

ROW 15 (117 - 128)

ROW 16 (129 - 135)

ROW 17 (135 - 141)

ROW 18 (142 - 151)

PROGRESSION OF A SLIDER CRANK

ROW 19 (152 - 161)

ROW 20 (162 - 166)

ROW 21 (167 - 175)

ROW 22 (176 - 186)

ROW 23 (187 - 196)

ROW 24 (197 - 207)

ROW 25 (208 - 210)

FREE VIBRATIONS

PROGRAM REGISTERS NEEDED: 41

ROW 1 (1 - 4)

ROW 2 (5 - 13)

ROW 3 (14 - 26)

ROW 4 (26 - 35)

ROW 5 (35 - 44)

ROW 6 (45 - 50)

ROW 7 (51 - 60)

ROW 8 (61 - 68)

ROW 9 (69 - 81)

ROW 10 (82 - 92)

ROW 11 (93 - 103)

ROW 12 (104 - 109)

ROW 13 (109 - 121)

RCW 14 (122 - 130)

ROW 15 (131 - 135)

ROW 16 (136 - 148)

ROW 17 (149 - 161)

ROW 18 (162 - 174)

ROW 19 (175 - 183)

ROW 20 (184 - 188)

ROW 21 (189 - 198)

ROW 22 (198 - 204)

INTERFERENCE FITS

PROGRAM REGISTERS NEEDED: 49

ROW 1 (1 - 3)

ROW 2 (4 - 7)

ROW 3 (8 - 14)

ROW 4 (14 - 17)

ROW 5 (17 - 23)

ROW 6 (23 - 28)

ROW 7 (29 - 32)

ROW 8 (33 - 40)

ROW 9 (40 - 45)

ROW 10 (46 - 50)

ROW 11 (50 - 58)

ROW 12 (59 - 71)

ROW 13 (72 - 84)

ROW 14 (85 - 94)

ROW 15 (95 - 101)

ROW 16 (101 - 105)

ROW 17 (106 - 111)

ROW 18 (112 - 120)

ROW 19 (120 - 125)

ROW 20 (126 - 131)

ROW 21 (132 - 138)

ROW 22 (138 - 145)

ROW 23 (146 - 150)

ROW 24 (151 - 156)

ROW 25 (157 - 166)

ROW 26 (167 - 174)

LINEAR OR ANGULAR DEFORMATION

PROGRAM REGISTERS NEEDED: 23

ROW 1 (1 - 3)

ROW 2 (4 - 10)

ROW 3 (10 - 18)

ROW 4 (19 - 26)

ROW 5 (27 - 34)

ROW 6 (34 - 37)

ROW 7 (37 - 46)

ROW 8 (46 - 54)

ROW 9 (55 - 63)

ROW 10 (64 - 70)

ROW 11 (70 - 81)

ROW 12 (82 - 89)

ROW 13 (90 - 92)

CONSTANT ACCELERATION

PROGRAM REGISTERS NEEDED: 22

ROW 1 (1 - 3)

ROW 2 (4 - 9)

ROW 3 (9 - 16)

ROW 4 (16 - 28)

ROW 5 (28 - 31)

ROW 6 (32 - 41)

ROW 7 (41 - 49)

ROW 8 (50 - 57)

ROW 9 (57 - 65)

ROW 10 (65 - 70)

ROW 11 (71 - 81)

ROW 12 (81 - 86)

NOTES

HEWLETT PACKARD

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*** Users' Library Solutions Books**

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

MECHANICAL ENGINEERING

GEAR FORCES
STRESS ON AN ELEMENT
EQUATIONS OF STATE
SODERBERG'S EQUATION FOR FATIGUE
SPRING CONSTANT
PROGRESSION OF A SLIDER CRANK
FREE VIBRATIONS
INTERFERENCE FITS
LINEAR OF ANGULAR DEFORMATION
CONSTANT ACCELERATION

HEWLETT PACKARD

00041-90090 Printed in U.S.A.